
When I first saw the 5-intersecting tetrahedron through my Honors class website, I found it to be beautiful and interesting. As I searched for project ideas, I kept coming back to it and knew I had to explore it further… 

THEORY AND CALCULATIONS
 I began with the five Platonic Solids: the cube, octahedron, icosohedron, dodecahedron, and tetrahedron. The properties that determine each platonic solid are that these convex polyhedra are composed of regular polygons of the same type and all of the corners on each solid are all the same. The Platonic Solids are also called the regular solids or regular polyhedra. Another important feature of the Platonic Solids is duality or reciprocity.  The principle of duality states that for every polyhedron there exists another polyhedron in which faces and vertices are in complementary locations. This duality pairs the Platonic Solids with one another. In order to determine the dual of a polyhedron, mark a point in the center of each face. Connect each point with lines, which will form the edges of the dual polyhedron. The reverse is also true. This reciprocity defines the duality of the two polyhedra. For example, by marking the center of the 6 faces of a cube and connecting the points, and octahedron is formed.
[image: image1.png]



Forming a Cube centered at the origin of three-dimensional space (0,0,0) with a distance of one unit to each point, the vertices are:
(-1,-1,-1)

(-1,-1,1)

(-1,1,-1)

(-1,1,1)

(1,-1,-1)

(1,-1,1)

(1,1,-1)

(1,1,1)

Therefore, the centers of each face would have the following centers, which are the vertices of the dual octahedron:
(0,0,-1)

(0,0,1)

(-1,0,0)

(1,0,0)

(0,-1,0)

(0,1,0)
Since the cube and the octahedron are duals, taking the centers of the faces of the octahedron would give us the another (smaller) cube, we must use another method to develop another Platonic Solid.


This can be done by expanding the points into equal line segments such that the faces of the octahedron become equilateral triangle faces. 

The length of this edge can be determined knowing this information. For example, the coordinates of the points I and J will be I = (t, 0,−1) and J = (−t, 0,−1), where t is a positive number between 0 and 1.
[image: image7.png]



Similarly, the coordinates of the points R and Q will be R = (1,− t,0) and L = (1, t ,0) for the same value of t. In order to determine the value of t so that when the points are joined each of the resulting triangles is equilateral, the distance formula is required. The distance formula for the length of a line segment between two points in the format (x,y) is 

. In an icosohedron, the distance from one vertex to an adjacent vertex would be the same throughout the polyhedron. One of the equilateral triangles in the icosohedron would be ∆IRQ. All of the sides would be equal with length 

. Therefore, 

 so by the binomial theoreom, 

 because length cannot be negative. Plugging in this value for t will determine the vertices of the regular icosohedron. 

The dodecahedron is the dual of the icosohedron. Therefore by determining the centers of the faces of the icosohedron, the vertices of the dodecahedron will be determined. In order to determine the centers of the faces, the sets of 3 vertices that make up each face had to be determined. This will be expained in more detail when it is done for the dodecahedron, but the 3 points are coplanar. These sets of vertices for the icosohedron are:

With this data, one can determine the coordinates of the centers of the faces in terms of t by determining the average of the 3 points. The data is shown below:

Using this data and substituting 

 gives us approximations of the points of all of the vertices of the dodecahedron.

Knowing these vertices, it was necessary to determine which points were collinear. This can be done using the distances between the vertices. The edges will be the shortest distances (0.7). Diagonals across the face will be the second distance (1.1). The third distance (1.5) will be the inside distance between the points that can be found by following one edge and one diagonal, or one diagonal then one edge. The fourth distance (1.9) is the inside distance between the points that can be found by traveling across one diagonal then another diagonal. I plotted the distances on a table and highlighted the shortest distance to find out which points shared an edge.

It is expected that each point will connect to 3 others, and those connections are shown in the table above. The relationships are shown in the table below:

Once the edges were determined, the points that made up each face of the dodecahedron had to be determined. Using the same method, choose a point. from that point, 2 of the three adjacent edges are chosen. From the point, there are 6 points that are of distance 1.1 (diagonal across a face). These are in pairs. Seeing a diagram of the faces laid out flat will help with this visualization.

Looking again at the distance table, we see that there are indeed, 6 points a distance of 1.1 from each vertex.

[image: image25.png]Distances between vertices of the Dodecahedron

Going from point 1, for example, we know that there are 3 edges, going to vertices 2, 4, and 14. looking at 2 and 4, there must be an edge from 2 and an edge from 4 that are both diagonals to 1, and are edges to each other. Looking at the table, 2 is an edge with 3 and 12. 4 is an edge with 5 and 18. 3 is an edge with 5 and 19, so the points that make up our first face are 1, 2, 3, 5, and 4. This is seen in the drawing below:

As you can see, 3 and 5 are diagonal to 1. This process is done for all the faces and the findings are listed in the table below:

It is helpful to use the drawing to determine the exact configurations when filling out the table.

Once all of the vertices that make up the faces are known, the dodecahedron can be plotted, and software used to display it. 
[image: image29.png]



We have now considered the cube, octahedron, icosohedron, and dodecahedron. The last remaining Platonic Solid is the tetrahedron. As we have previously discussed, if we place a point on the center of the faces of the dodecahedron and connect those points we will have a smaller icosohedron. In order to develop a tetrahedron, we take 4 equidistant corners of the dodecahedron and connect them with lines, the result would be a pyramid (a tetrahedron) inscribed in the dodecahedron.
This is shown below:
[image: image30.png]



This tetrahedron has 4 vertices and the dodecahedron has 20 vertices. We could inscribe 5 distinct tetrahedra inside a dodecahedron. This is shown here, first without the dodecahedron frame, then with it:

This beautiful and symmetrical mathematical object is called the five-intersecting tetrahedral.
MODEL DEVELOPMENT


The model that I built for the class was developed from Make Shapes
. This model, which will be shown in class, was scanned into the computer. I then drew one section of the model using Printmaster 11 software, copied that image 3 times, and converted the document to a picture. I moved the picture into Micorsoft Word so that I could resize it until I was happy with the size. I separated the corners so that I could print each tetrahedron on different color card stock. There were 4 images to a page because there are four corners in a tetrahedron. I resized the images because I wanted a larger model. Here is a picture of one of the sections:
[image: image36.png]



POV-Ray Code

The POV-Ray
 code was developed using the methods and mathematics described above in “Theory and Calculations”. The code used to develop the Cube -> Octahedron -> Icosohedron pictures
 can be found on the HON301 course website. The code used to develop the Dodecahedron, the 5-intersecting tetrahedral, and the movie is below:
//Dodecahedron DEFINED BY THE COORDINATES OF ITS VERTICES AND POLYGONS

//Files with predefined colors and textures

#include "colors.inc"

#include "glass.inc"

#include "golds.inc"

#include "metals.inc"

#include "stones.inc"

#include "woods.inc"

//Place the camera

camera {

  sky <0,0,1>          //Don't change this

  direction <-1,0,0>   //Don't change this  

  right <-4/3,0,0>     //Don't change this

  location  <30,10,10>  //Camera location

  look_at   <0,0,0>    //Where camera is pointing

  angle 5       //Angle of the view--increase to see more, decrease to see less

} 

//Ambient light to "brighten up" darker pictures

global_settings { ambient_light White*2 } 

//Place a light--you can have more than one! 

light_source {<10,5,10> color White*2 }

light_source {<10,-5,10> color White*2 }

light_source {<-10,5,10> color White*2 }

//Set a background color

background { color VLightGrey } 

//Construct a docecahedron by describing its polygons

//The 20 vertices

#declare tt=(sqrt(5)-1)/2;

#declare w1=<tt,0,-1>;

#declare w2=<-tt,0,-1>;

#declare w3=<0,-1,tt>;

#declare w4=<0,-1,-tt>;

#declare w5=<-1,tt,0>;

#declare w6=<-1,-tt,0>;

#declare w7=<0,1,tt>;

#declare w8=<0,1,-tt>;

#declare w9=<1,tt,0>;

#declare w10=<1,-tt,0>;

#declare w11=<tt,0,1>;

#declare w12=<-tt,0,1>;

#declare v1 = (w1+w8+w9)/3;

#declare v2 = (w1+w9+w10)/3;

#declare v3 = (w1+w4+w10)/3;

#declare v4 = (w1+w2+w8)/3;

#declare v5 = (w1+w2+w4)/3;

#declare v6 = (w7+w11+w12)/3;

#declare v7 = (w5+w7+w12)/3;

#declare v8 = (w5+w6+w12)/3;

#declare v9 = (w3+w6+w12)/3;

#declare v10 = (w3+w11+w12)/3;

#declare v11 = (w3+w10+w11)/3;

#declare v12 = (w9+w10+w11)/3;

#declare v13 = (w7+w9+w11)/3;

#declare v14 = (w7+w8+w9)/3;

#declare v15 = (w5+w7+w8)/3;

#declare v16 = (w2+w5+w6)/3;

#declare v17 = (w2+w4+w6)/3;

#declare v18 = (w2+w5+w8)/3;

#declare v19 = (w3+w4+w10)/3;

#declare v20 = (w3+w4+w6)/3; 

//The 12 pentagonal faces

#declare fd1 = polygon  { 6, v1, v4,  v18, v15, v14, v1 texture{pigment {color rgbf <0,0,1,0>}}};

#declare fd2 = polygon  { 6, v1, v2,  v3,  v5,  v4,  v1 texture{pigment {color rgbf <0,1,0,0>}}};

#declare fd3 = polygon  { 6, v1, v2,  v12, v13, v14, v1 texture{pigment {color rgbf <1,0,0,0>}}};

#declare fd4 = polygon  { 6, v2, v3,  v19, v11, v12, v2 texture{pigment {color rgbf <0,0,.5,0>}}};

#declare fd5 = polygon  { 6, v3, v5,  v17, v20, v19, v3 texture{pigment {color rgbf <0,.5,0,0>}}};

#declare fd6 = polygon  { 6, v4, v5,  v17, v16, v18, v4 texture{pigment {color rgbf <.5,0,0,0>}}};

#declare fd7 = polygon  { 6, v7, v8,  v16, v18, v15, v7 texture{pigment {color rgbf <0,.5,1,0>}}};

#declare fd8 = polygon  { 6, v6, v10, v11, v12, v13, v6 texture{pigment {color rgbf <.5,0,1,0>}}};

#declare fd9 = polygon  { 6, v6, v7,  v15, v14, v13, v6 texture{pigment {color rgbf <.5,1,0,0>}}};

#declare fd10 = polygon { 6, v6, v7,  v8,  v9,  v10, v6 texture{pigment {color rgbf <1,0,.5,0>}}};

#declare fd11 = polygon { 6, v8, v9,  v20, v17, v16, v8 texture{pigment {color rgbf <1,.5,0,0>}}};

#declare fd12 = polygon { 6, v9, v10, v11, v19, v20, v9 texture{pigment {color rgbf <0,.5,1,0>}}};

//Unite the faces to form the dodecahedron

#declare mydodecahedron = object { union {

  object{fd1}

  object{fd2} 

  object{fd3} 

  object{fd4} 

  object{fd5} 

  object{fd6}

  object{fd7} 

  object{fd8} 

  object{fd9} 

  object{fd10}

  object{fd11}

  object{fd12}

  }

  }

//List the defined object(s) to be displayed

mydodecahedron
//FIVE INTERSECTING TETRAHEDRA DEFINED BY THE COORDINATES OF ITS VERTICES AND POLYGONS

//Files with predefined colors and textures

#include "colors.inc"

#include "glass.inc"

#include "golds.inc"

#include "metals.inc"

#include "stones.inc"

#include "woods.inc"

//Place the camera

camera {

  sky <0,0,1>          //Don't change this

  direction <-1,0,0>   //Don't change this  

  right <-4/3,0,0>     //Don't change this

  location  <30,10,10>  //Camera location

  look_at   <0,0,0>    //Where camera is pointing

  angle 5       //Angle of the view--increase to see more, decrease to see less

}

//Ambient light to "brighten up" darker pictures

global_settings { ambient_light White*3 }

//Place a light--you can have more than one! 

light_source {<10,5,10> color White*2 }

light_source {<10,-5,10> color White*2 }

light_source {<-10,5,-10> color White*2 }

//Set a background color

background {color VLightGrey}

//The 20 vertices

#declare v1 = <((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3>;

#declare v2 = <((sqrt(5)-1)/2+2)/3,0,-1/3>;

#declare v3 = <((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3>;

#declare v4 = <0,1/3,-((sqrt(5)-1)/2+2)/3>;

#declare v5 = <0,-1/3,-((sqrt(5)-1)/2+2)/3>;

#declare v6 = <0,1/3,((sqrt(5)-1)/2+2)/3>;

#declare v7 = <-((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3>;

#declare v8 = <-((sqrt(5)-1)/2+2)/3,0,1/3>;

#declare v9 = <-((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3>;

#declare v10 = <0,-1/3,((sqrt(5)-1)/2+2)/3>;

#declare v11 = <((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3>;

#declare v12 = <((sqrt(5)-1)/2+2)/3,0,1/3>;

#declare v13 = <((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3>;

#declare v14 = <1/3,((sqrt(5)-1)/2+2)/3,0>;

#declare v15 = <-1/3,((sqrt(5)-1)/2+2)/3,0>;

#declare v16 = <-((sqrt(5)-1)/2+2)/3,0,-1/3>;

#declare v17 = <-((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3>;

#declare v18 = <-((sqrt(5)-1)/2+1)/3,((sqrt(5)-1)/2+1)/3,-((sqrt(5)-1)/2+1)/3>;

#declare v19 = <1/3,-((sqrt(5)-1)/2+2)/3,0>;

#declare v20 = <-1/3,-((sqrt(5)-1)/2+2)/3,0>;

//Construct first tetrahedron by describing its polygons

//---------------------------------------------------------------

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f1_1 = polygon { 4, v1, v6, v16, v1 };

#declare f1_2 = polygon { 4, v1, v16, v19, v1 };

#declare f1_3 = polygon { 4, v1, v19, v6, v1 };

#declare f1_4 = polygon { 4, v6, v16, v19, v6 };

//Define the pyramid to be the union of its faces.

#declare mytetra1 = object { union {

  object{f1_1}

  object{f1_2} 

  object{f1_3} 

  object{f1_4}

  }

 pigment { Magenta }

  };       

//Construct second tetrahedron by describing its polygons

//---------------------------------------------------------------

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f2_1 = polygon { 4, v2, v10, v15, v2 };

#declare f2_2 = polygon { 4, v2, v15, v17, v2 };

#declare f2_3 = polygon { 4, v2, v17, v10, v2 };

#declare f2_4 = polygon { 4, v10, v15, v17, v10 };

//Define the pyramid to be the union of its faces.

#declare mytetra2 = object { union {

  object{f2_1}

  object{f2_2} 

  object{f2_3} 

  object{f2_4}

  }

  pigment { NeonBlue }

  };  

//Construct third tetrahedron by describing its polygons

//---------------------------------------------------------------

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f3_1 = polygon { 4, v3, v9, v13, v3 };

#declare f3_2 = polygon { 4, v3, v13, v18, v3 };

#declare f3_3 = polygon { 4, v3, v18, v9, v3 };

#declare f3_4 = polygon { 4, v9, v13, v18, v9 };

//Define the pyramid to be the union of its faces.

#declare mytetra3 = object { union {

  object{f3_1}

  object{f3_2} 

  object{f3_3} 

  object{f3_4}

  }

  pigment { White } 

  };

//Construct fourth tetrahedron by describing its polygons

//---------------------------------------------------------------

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f4_1 = polygon { 4, v4, v7, v12, v4 };

#declare f4_2 = polygon { 4, v4, v12, v20, v4 };

#declare f4_3 = polygon { 4, v4, v20, v7, v4 };

#declare f4_4 = polygon { 4, v7, v12, v20, v7 };

//Define the pyramid to be the union of its faces.

#declare mytetra4 = object { union {

  object{f4_1}

  object{f4_2} 

  object{f4_3} 

  object{f4_4}

  }

  pigment { MandarinOrange } 

  };

//Construct fifth tetrahedron by describing its polygons

//---------------------------------------------------------------

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f5_1 = polygon { 4, v5, v8, v11, v5 };

#declare f5_2 = polygon { 4, v5, v11, v14, v5 };

#declare f5_3 = polygon { 4, v5, v14, v8, v5 };

#declare f5_4 = polygon { 4, v8, v11, v14, v8 };

//Define the pyramid to be the union of its faces.

#declare mytetra5 = object { union {

  object{f5_1}

  object{f5_2} 

  object{f5_3} 

  object{f5_4}

  }

  pigment { Yellow } 

  };

//Intersect the 5 tetrahedron

merge{

  object{mytetra1}

  object{mytetra2} 

  object{mytetra3} 

  object{mytetra4}  

  object{mytetra5}

  }

//FIVE TETRAHEDRA MOVIE

//ANIMATE WITH CLOCK FROM 0 TO 7

//Files with predefined colors and textures

#include "colors.inc"

#include "glass.inc"

#include "golds.inc"

#include "metals.inc"

#include "stones.inc"

#include "woods.inc"   

global_settings { max_trace_level 5 }  

//Place the camera

camera {

  sky <0,0,1>          //Don't change this

  direction <-1,0,0>   //Don't change this  

  right <-4/3,0,0>     //Don't change this

  location  <10,10,10>  //Camera location

  look_at   <0,0,0>    //Where camera is pointing

  angle 15             //Angle of the view--increase to see more, decrease to see less

}             

global_settings { ambient_light White*2 } //Ambient light to "brighten up" darker pictures

//Place a light--you can have more than one! 

light_source {<10,5,10> color White*2 }

light_source {<10,-5,10> color White*2 }

//Set a background color

background { color VLightGrey }   

//The Icosahedron (adapted from Laura Berry)

//Vertices are defined with tt so that the icosohedron is developed from an octahedron

#macro myicosahedron(tt)

//The 12 vertices

#declare p1 = <tt,0,-1>;

#declare p2 = <-tt,0,-1>;

#declare p3 = <0,-1,tt>;

#declare p4 = <0,-1,-tt>;

#declare p5 = <-1,tt,0>;

#declare p6 = <-1,-tt,0>;

#declare p7 = <0,1,tt>;

#declare p8 = <0,1,-tt>;

#declare p9 = <1,tt,0>;

#declare p10 = <1,-tt,0>;

#declare p11 = <tt,0,1>;

#declare p12 = <-tt,0,1>; 

//The 20 triangular faces

#declare f1 = polygon { 4, p1, p8, p9, p1};

#declare f2 = polygon { 4, p1, p10, p9, p1};

#declare f3 = polygon { 4, p1, p10, p4, p1};

#declare f4 = polygon { 4, p1, p2, p8, p1};

#declare f5 = polygon { 4, p1, p2, p4, p1};

#declare f6 = polygon { 4, p12, p7, p11, p12};

#declare f7 = polygon { 4, p12, p7, p5, p12};

#declare f8 = polygon { 4, p12, p5, p6, p12};

#declare f9 = polygon { 4, p12, p6, p3, p12};

#declare f10 = polygon { 4, p12, p3, p11, p12};

#declare f11 = polygon { 4, p3, p10, p11, p3};

#declare f12 = polygon { 4, p9, p10, p11, p9};

#declare f13 = polygon { 4, p7, p9, p11, p7};

#declare f14 = polygon { 4, p7, p8, p9, p7};

#declare f15 = polygon { 4, p5, p8, p7, p5};

#declare f16 = polygon { 4, p5, p2, p6, p5};

#declare f17 = polygon { 4, p2, p6, p4, p2};

#declare f18 = polygon { 4, p2, p5, p8, p2};

#declare f19 = polygon { 4, p3, p4, p10, p3};

#declare f20 = polygon { 4, p3, p4, p6, p3};

//Unite the faces to form the icosahedron

object { union {

  object{f1}

  object{f2} 

  object{f3} 

  object{f4} 

  object{f5} 

  object{f6}

  object{f7} 

  object{f8} 

  object{f9} 

  object{f10}

  object{f11}

  object{f12} 

  object{f13} 

  object{f14} 

  object{f15} 

  object{f16}

  object{f17} 

  object{f18} 

  object{f19} 

  object{f20}

  }

  }

#end

//The Dodecahedron (adapted from the icosohedron using the center of each icosohedron face as a vertex

#macro mydodecahedron(tt)

#declare p1 = <tt,0,-1>;

#declare p2 = <-tt,0,-1>;

#declare p3 = <0,-1,tt>;

#declare p4 = <0,-1,-tt>;

#declare p5 = <-1,tt,0>;

#declare p6 = <-1,-tt,0>;

#declare p7 = <0,1,tt>;

#declare p8 = <0,1,-tt>;

#declare p9 = <1,tt,0>;

#declare p10 = <1,-tt,0>;

#declare p11 = <tt,0,1>;

#declare p12 = <-tt,0,1>; 

//The 20 vertices

#declare v1 = (p1+p8+p9)/3;

#declare v2 = (p1+p9+p10)/3;

#declare v3 = (p1+p4+p10)/3;

#declare v4 = (p1+p2+p8)/3;

#declare v5 = (p1+p2+p4)/3;

#declare v6 = (p7+p11+p12)/3;

#declare v7 = (p5+p7+p12)/3;

#declare v8 = (p5+p6+p12)/3;

#declare v9 = (p3+p6+p12)/3;

#declare v10 = (p3+p11+p12)/3;

#declare v11 = (p3+p10+p11)/3;

#declare v12 = (p9+p10+p11)/3;

#declare v13 = (p7+p9+p11)/3;

#declare v14 = (p7+p8+p9)/3;

#declare v15 = (p5+p7+p8)/3;

#declare v16 = (p2+p5+p6)/3;

#declare v17 = (p2+p4+p6)/3;

#declare v18 = (p2+p5+p8)/3;

#declare v19 = (p3+p4+p10)/3;

#declare v20 = (p3+p4+p6)/3;

//The 12 pentagonal faces

#declare fd1 = polygon  { 6, v1, v4,  v18, v15, v14, v1 };

#declare fd2 = polygon  { 6, v1, v2,  v3,  v5,  v4,  v1 };

#declare fd3 = polygon  { 6, v1, v2,  v12, v13, v14, v1 };

#declare fd4 = polygon  { 6, v2, v3,  v19, v11, v12, v2 };

#declare fd5 = polygon  { 6, v3, v5,  v17, v20, v19, v3 };

#declare fd6 = polygon  { 6, v4, v5,  v17, v16, v18, v4 };

#declare fd7 = polygon  { 6, v7, v8,  v16, v18, v15, v7 };

#declare fd8 = polygon  { 6, v6, v10, v11, v12, v13, v6 };

#declare fd9 = polygon  { 6, v6, v7,  v15, v14, v13, v6 };

#declare fd10 = polygon { 6, v6, v7,  v8,  v9,  v10, v6 };

#declare fd11 = polygon { 6, v8, v9,  v20, v17, v16, v8 };

#declare fd12 = polygon { 6, v9, v10, v11, v19, v20, v9 }; 

   //Unite the faces to form the dodecahedron

object { union {

  object{fd1}

  object{fd2} 

  object{fd3} 

  object{fd4} 

  object{fd5} 

  object{fd6}

  object{fd7} 

  object{fd8} 

  object{fd9} 

  object{fd10}

  object{fd11}

  object{fd12}

  }

  }

#end  

//The 5-Intersecting Tetrahedron (made by creating each tetrahedron from polygons made from specific vertices of the dodecahedron

#declare tt=(sqrt(5)-1)/2;

#declare p1 = <tt,0,-1>;

#declare p2 = <-tt,0,-1>;

#declare p3 = <0,-1,tt>;

#declare p4 = <0,-1,-tt>;

#declare p5 = <-1,tt,0>;

#declare p6 = <-1,-tt,0>;

#declare p7 = <0,1,tt>;

#declare p8 = <0,1,-tt>;

#declare p9 = <1,tt,0>;

#declare p10 = <1,-tt,0>;

#declare p11 = <tt,0,1>;

#declare p12 = <-tt,0,1>;

//The 20 vertices

#declare v1 = (p1+p8+p9)/3;

#declare v2 = (p1+p9+p10)/3;

#declare v3 = (p1+p4+p10)/3;

#declare v4 = (p1+p2+p8)/3;

#declare v5 = (p1+p2+p4)/3;

#declare v6 = (p7+p11+p12)/3;

#declare v7 = (p5+p7+p12)/3;

#declare v8 = (p5+p6+p12)/3;

#declare v9 = (p3+p6+p12)/3;

#declare v10 = (p3+p11+p12)/3;

#declare v11 = (p3+p10+p11)/3;

#declare v12 = (p9+p10+p11)/3;

#declare v13 = (p7+p9+p11)/3;

#declare v14 = (p7+p8+p9)/3;

#declare v15 = (p5+p7+p8)/3;

#declare v16 = (p2+p5+p6)/3;

#declare v17 = (p2+p4+p6)/3;

#declare v18 = (p2+p5+p8)/3;

#declare v19 = (p3+p4+p10)/3;

#declare v20 = (p3+p4+p6)/3;

//Construct first tetrahedron by describing its polygons

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f1_1 = polygon { 4, v1, v6, v16, v1 };

#declare f1_2 = polygon { 4, v1, v16, v19, v1 };

#declare f1_3 = polygon { 4, v1, v19, v6, v1 };

#declare f1_4 = polygon { 4, v6, v16, v19, v6 }; 

//Define the pyramid to be the union of its faces.

#declare mytetra1 = object { union {

  object{f1_1}

  object{f1_2} 

  object{f1_3} 

  object{f1_4}

  }

 pigment { Magenta }

  };       

//---------------------------------------------------------------

//Construct second tetrahedron by describing its polygons

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f2_1 = polygon { 4, v2, v10, v15, v2 };

#declare f2_2 = polygon { 4, v2, v15, v17, v2 };

#declare f2_3 = polygon { 4, v2, v17, v10, v2 };

#declare f2_4 = polygon { 4, v10, v15, v17, v10 };

//Define the pyramid to be the union of its faces.

#declare mytetra2 = object { union {

  object{f2_1}

  object{f2_2} 

  object{f2_3} 

  object{f2_4}

  }

  pigment { NeonBlue }

  };  

//---------------------------------------------------------------

//Construct third tetrahedron by describing its polygons

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f3_1 = polygon { 4, v3, v9, v13, v3 };

#declare f3_2 = polygon { 4, v3, v13, v18, v3 };

#declare f3_3 = polygon { 4, v3, v18, v9, v3 };

#declare f3_4 = polygon { 4, v9, v13, v18, v9 };

//Define the pyramid to be the union of its faces.

#declare mytetra3 = object { union {

  object{f3_1}

  object{f3_2} 

  object{f3_3} 

  object{f3_4}

  }

  pigment { White } 

  };

//---------------------------------------------------------------

//Construct fourth tetrahedron by describing its polygons

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f4_1 = polygon { 4, v4, v7, v12, v4 };

#declare f4_2 = polygon { 4, v4, v12, v20, v4 };

#declare f4_3 = polygon { 4, v4, v20, v7, v4 };

#declare f4_4 = polygon { 4, v7, v12, v20, v7 };

//Define the pyramid to be the union of its faces.

#declare mytetra4 = object { union {

  object{f4_1}

  object{f4_2} 

  object{f4_3} 

  object{f4_4}

  }

  pigment { MandarinOrange } 

  };

//---------------------------------------------------------------

//Construct fifth tetrahedron by describing its polygons

//Define the polygons for each of the five sides.

//Notice that the last point must be a repetition of the first.

//The first number indicates how many points will be listed.

#declare f5_1 = polygon { 4, v5, v8, v11, v5 };

#declare f5_2 = polygon { 4, v5, v11, v14, v5 };

#declare f5_3 = polygon { 4, v5, v14, v8, v5 };

#declare f5_4 = polygon { 4, v8, v11, v14, v8 };

//Define the pyramid to be the union of its faces.

#declare mytetra5 = object { union {

  object{f5_1}

  object{f5_2} 

  object{f5_3} 

  object{f5_4}

  }

  pigment { Yellow } 

  };

//---------------------------------------------------------------

//Intersect the 5 tetrahedron

#declare myonetetra=

merge{

  object{mytetra1}

  }   

#declare mytwotetra=

merge{

  object{mytetra1}

  object{mytetra2} 

  }   

#declare mythreetetra=

merge{

  object{mytetra1}

  object{mytetra2} 

  object{mytetra3} 

  }   

#declare myfourtetra=

merge{

  object{mytetra1}

  object{mytetra2} 

  object{mytetra3} 

  object{mytetra4}  

  }  

#declare myfivetetra=

merge{

  object{mytetra1}

  object{mytetra2} 

  object{mytetra3} 

  object{mytetra4}  

  object{mytetra5}

  }   

#declare rate=90;

#declare f=.7;

#switch(clock)

#range(0,1)

#declare s=clock;

#declare tt=0;

union{

box{<-1,-1,-1>,<1,1,1> texture{ pigment{ color rgbf <1,s,s,s*f>}}}

object{myicosahedron(0.0001) texture{pigment {color rgbf <0,1,0,0>}}}

rotate<0,0,clock*rate>

}     

#break

#range(1.0001,1.5)

#declare s=clock-1;

#declare r=s*(sqrt(5)-1)/2;

union{

box{<-1,-1,-1>,<1,1,1> texture{ pigment{ color rgbf <1,1,1,f>}}}

object{myicosahedron(r) texture{pigment {color rgbf <0,1,0,0>}}}

rotate<0,0,clock*rate>

}      

#break

#range(1.5001,2)

#declare s=clock-1;

#declare r=(sqrt(5)-1)/2;

union{

object{myicosahedron(r) texture{pigment {color rgbf <s,1,s,s*f>}}}

object{mydodecahedron(r) texture{pigment{color rgbf <1,0,0,0>}}}

rotate<0,0,clock*rate>

}     

#break

#range(2.0001,3)

#declare r=(sqrt(5)-1)/2;

object{mydodecahedron(r) texture{pigment{color rgbf <1,0,0,0>}} rotate <0,0,clock*rate>} 

#break

#range(3.0001,4)

#declare s=clock-3;

#declare r=(sqrt(5)-1)/2;   

union{

object{mydodecahedron(r) texture{pigment {color rgbf <1,s,s,s*f>}}}

object{myonetetra texture{pigment{color Gold*2}}}

rotate<0,0,clock*rate>

} 

#break

#range(4.0001,4.5)

#declare r=(sqrt(5)-1)/2;

object{mytwotetra texture{pigment{color Gold*2}} rotate <0,0,clock*rate>}  

#break

#range(4.5001,5)

#declare r=(sqrt(5)-1)/2;

object{mythreetetra texture{pigment{color Gold*2}} rotate <0,0,clock*rate>}

#break

#range(5.0001,5.5)

#declare r=(sqrt(5)-1)/2;

object{myfourtetra texture{pigment{color Gold*2}} rotate <0,0,clock*rate>}

#break

#range(5.5001,6)

#declare r=(sqrt(5)-1)/2;

object{myfivetetra texture{pigment{color Gold*2}} rotate <0,0,clock*rate>}

#break

#range(6.0001,7)

#declare r=(sqrt(5)-1)/2;   

union{

object{mydodecahedron(r) texture{pigment {color rgbf <1,1,1,0.7>}}}

object{myfivetetra texture{pigment{color Gold*2}}}

rotate<0,0,clock*rate>

} 

#break

#end




References:





Most information taken from knowledge learned in University of Kentucky HON 301 Visualizing Mathematics Class: Notes located at http://www.ms.uky.edu/~lee/visual05/visual05.html





� Make Shapes Series No. 2: 8 mathematical models to cut out, glue and decorate, Gerald Jenkins and Anne Wild, Tarquin Productions 1990, Original Edition 1978, ISBN 0 906212 01 4


� POV-Ray is a free Ray tracing software available at http://www.povray.org/


� Taken from file developed by Dr. Lee published on http://www.ms.uky.edu/~lee/visual05/povray/povray.html

